LISTA D

Cássio Murakami 10773798

Exercício 1: Faça as modificações adequadas para se poder desenhar e comparar os gráficos da resposta do sistema não linear e linear. Faça as simulações dos sistemas linear e não linear considerando que o reservatório parte do nível h = 2 m, mas com vazão de entrada nula. Compare as respostas.

- Estudando o seguinte sistema hidráulico:



A equação diferencial não-linear que modela o sistema é a seguinte:

$$\dot{h} = \left(-\sqrt{\frac{\rho g h}{R}} + Q_e \right) \frac{1}{S}$$

Linearizando o sistema, a seguinte equação revela a evolução temporal da variação da altura do reservatório em relação a altura em regime:

$$\dot{x} = -\frac{1}{2S}\sqrt{\frac{\rho g}{R h_o}}\, x + \frac{1}{S} u$$

$$y = +1x + 0u$$

Utilizando o software *Scilab 6.1.0,* foi elaborado um código para esboçar e comparar a solução numérica obtida a partir do modelo não-linear e a solução obtida do sistema linearizado.

Os seguintes parâmetros foram considerados para a simulação:

$S = 10\ m^2$

$\rho = 1000\ kg/m^3$

$g = 10\ m/s^2$

$h_o = 2\ m$ (Nível do reservatório em regime)

$h_i = 0{,}1\ m$ (Nível adicional desejado)

Dessa maneira, solucionando o problema não-linear e linearizado os seguintes resultados são obtidos:



➢ O seguinte código foi elaborado para a solução do presente problema:

```
clear all

// Definir parametros:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
Qei=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada

// Definir o sistema linear usando o comando syslin:
```

```
A=(-1/(2*S))*sqrt(rho*g/(R*ho));
B=1/S;
C=1;
D=0;
tanque=syslin('c',A,B,C,D);

// Definir a condicao inicial:
x0=0; // [m] desvio inicial do nivel em relação ao equilibrio
h0 = ho + x0; // [m] Altura inicial do tanque

// Definir o vetor de instantes de tempo:
t=0:10:40000;

// Definir o vetor de entradas:
u=Qei*ones(t);

// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);

Qei=sqrt(rho*g*(ho+hi)/R);

//Solução numérica do problema não-linear:
funcprot(0);
  function fun=num_sol(t, h)
    fun =(-sqrt(rho*g*h/R)+Qei)*(1/S)
  endfunction

y_num = ode(h0,t(1),t,num_sol);

// Plot dos resultados
scf(1);
plot(t,y);
plot(t,y_num - ho,'r');
xtitle("Resposta do tanque","Tempo t [s]","Variacao de nivel [m]");
legend(['Solução linear';'Solução não-linear'])
```

Exercício 2: Obtenha o modelo matemático do circuito elétrico mostrado abaixo e compare com o modelo linear do sistema com um reservatório. Faça simulações e compare qualitativamente com os resultados do exercício 1 (sistema linear).



A partir da lei das malhas, e das equações constitutivas de cada componente do circuito elétrico, é possível obter a evolução temporal da tensão no capacitor:

$$e_i - Ri - \frac{1}{C}\int i\,dt = 0$$

Realizando a seguinte mudança de variável: $i = \frac{dy}{dt}$ (y é a carga elétrica)

$$e_i - R\dot{y} - \frac{1}{C}y = 0 \;\rightarrow\; \dot{y} + \frac{1}{RC}y = \frac{e_i}{R}$$

Será considerada a seguinte condição inicial $y(0) = 0$. Tal escolha é justificada para se comparar qualitativamente com a equação linear do sistema linear dos reservatório.

Observando as equações resultantes do sistema elétrico e da linearizada do sistema hidráulico, é possível notar uma semelhança entre a forma (Equação diferencial de primeira ordem com forçante). Dessa maneira, escolhendo os valores numéricos correspondentes para os sistemas serem coincidentes:

$$\frac{1}{RC} = \frac{1}{2S}\sqrt{\frac{\rho g}{Rh_o}}$$

$$\frac{e_i}{R} = \frac{u}{S}$$

Se as relações acima forem satisfeitas, o resultado do sistema elétrico será idêntico ao sistema do reservatório.



Tarefa de casa: Usando a abordagem vista nestes exemplos, faça a simulação do sistema com dois reservatórios, supondo o modelo linear:



A partir da lista C, foi deduzida a linearização do sistema de equações diferenciais que correspondem as alturas dos reservatórios:

$$\dot{h_1} = -\frac{1}{2S_2}\frac{\rho g}{Q_{ei}R_a}(h_1 - h_2) + \frac{Q_{ei}}{2S_1}$$

$$\dot{h_2} = \frac{1}{2S_2}\frac{\rho g}{Q_{ei}R_a}(h_1 - h_2) - \frac{1}{2S_2}\frac{\rho g}{Q_{ei}R_S}h_2$$

Integrando numericamente com os mesmos parâmetros utilizados no exercício anterior nos dois reservatórios, o seguinte resultado é obtido:



```
clear all

// Definir parametros:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
//Qei=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada
Qei=sqrt(rho*g*(ho+hi)/R)

//Vetor dos instantes de tempo:
t = 0:10:40000;

//Alturas iniciais dos resevatórios:
h01 = ho;
h02 = ho;

function dy=h_num(t, y)
   dy(1) = -(1/(2*S))*((rho*g)/(Qei*R))*(y(1)-y(2)) + Qei/(2*S)
   dy(2) = (1/(2*S))*((rho*g)/(Qei*R))*(y(1)-y(2)) - (1/(2*S))*((rho*g)/(Qei*R))*y(2)
endfunction

//Integração numérica da equação linear
sol = ode([ho;ho],0,t,h_num);
h1 = sol(1,:);
h2 = sol(2,:);

//Plot das soluções:
scf(1)
xtitle("Nível dos reservatórios")
xlabel("Tempo [s]")
ylabel("Altura do reservatório [m]")
plot(t,h1,'r')
plot(t,h2,'g')
legend(['Reservatório 1','Rervatório 2'])
```